

Excel 函数用于解决水文流量计算中 尾数取舍问题的简单方法

李 伟, 李 弘, 彭学成

(四川省成都水文水资源勘测局, 四川 成都 611130)

摘要: 为提高水文站实测流量计算的工作效率, 研究拟采用 Excel 软件对实测流量进行计算。对于采用 Excel 软件进行流量计算过程中遇到的尾数取舍问题, 考虑采用 Excel 函数进行解决。经过研究, 仅仅使用了 3 个 Excel 函数就成功地解决了流量计算过程中遇到的尾数取舍问题, 使流量计算过程中的尾数取舍结果均符合水文规范的要求。从而大大地简化了实测流量的计算过程, 节约了实测流量计算时间。该方法简单易学, 便于推广, 从而能提高水文流量测验的工作效率。

关键词: 流量计算; Excel 函数; 尾数取舍

中图分类号: P333

文献标识码: B

文章编号: 1007-7839 (2018) 08-0059-04

Study on the simple method of Excel function to solve the problem of mantissa option in the calculation of hydrological flow

LI Wei, LI Hong, PENG Xuecheng

(Chengdu Hydrology and Water Resources Investigation Bureau of Sichuan Province, Chengdu 611130, Sichuan)

Abstract: In order to improve the calculation efficiency of the measured flow of hydrological stations, the Excel software was used to calculate the measured discharge. It's considered to use Excel function to solve the mantissa option problem encountered in the process of flow calculation using Excel software. After the study, only using 3 Excel functions could successfully solve the mantissa option problem in the flow calculation, and the results were all in accordance with the requirements of the hydrological norms. Thus, the calculation process of measured flow was greatly simplified, and the computation time of measured flow was saved. This method is simple to learn and easy to popularize, so as to improve the efficiency of hydrological discharge test.

Key words: flow calculation; Excel function; mantissa option

0 引言

流量计算是水文站水文测验资料收集中非常重要的一项基本工作。传统方法进行流量计算需用计算器一步一步计算平均水深、部分断面面积、

平均流速、部分流量, 然后汇总统计出实测流量。中间还需要对各步计算结果进行有效位数和尾数人工取舍。故 1 份实测流量的整个计算过程是非常复杂和耗时的。在电脑普及化程度非常高的今天, 各个水文站均配置了电脑, 故笔者考虑采用

收稿日期: 2018-03-16

作者简介: 李伟 (1982—), 男, 工程师, 硕士, 主要从事水文水资源和水土保持研究工作。

Excel 软件对实测流量进行计算。Excel 作为一个常用的计算机办公软件,其计算的功能非常强大,可以为我们进行实测流量计算提供简便的方法。但在应用时我们发现了一个问题,就是 Excel 软件的尾数取舍是常规的四舍五入,而并非水文规范要求的尾数取舍方式。为了方便实测流量计算,笔者对 Excel 软件进行水文流量计算中的尾数取舍问题进行研究和试验,找到一种简单便捷的方法。

1 流量计算的尾数取舍方式

规范文件《数值修约规则与极限数值的表示和判定》(GB/T8170-2008)^[1]和《水文资料整编规范》(SL247-2012)对流量计算的尾数取舍方式进行了规范。尾数取舍方式^[2]如下:“取用精度位数后一位数字,采用‘四舍六入’方法取舍,即取用精度位数后一位数字小于五者则舍,大于五者则入,等于五时若其后有非零尾数仍入,无非零尾数则视取用的末位数字的奇偶取舍,为奇则入,为偶则舍。”

2 使用 Excel 函数实现流量计算的尾数取舍

2.1 采用 Excel 函数

ROUND 函数和 MOD 函数,关于这 2 个函数的介绍和用法可以参看本文参考文献^[3]或者 Excel 软件的帮助文档。

2.2 实现流量计算的尾数取舍

由前节所述水文流量“四舍六入”取舍方法可知,这种取舍方法与 Excel 软件中 ROUND 函数采用的常规的“四舍五入”取舍方法是有差异的。差异如表 1 所示。

由上图分析得知 2 种取舍方法只有在一种情

况下才存在取舍差异。即取用精度位数后一位数字等于 5,且其后无非零尾数,且取用的末位数字为偶数时,水文流量“四舍六入”法为“舍”,Excel“四舍五入”法为“入”。由此分析,保留两位小数的水文流量“四舍六入”的取舍方法可采用以下函数式实现。

$$=ROUND(A2, n) - (MOD(A2 * 10^n, 2) = 0.5) / 10^n \quad (1)$$

式中, n 为保留小数位数,与 ROUND 函数参数定义相同,比如 1 为保留 1 位小数,0 位为保留个位数, -1 为保留到十位数等。

本文以保留 2 位小数的取舍示例如图 2。

	A	B	C	D	E
1	原始数据	水文流量“四舍六入”法	Excel“四舍五入”法		
2	1.1141	1.11	1.11		
3	1.1262	1.13	1.13		
4	1.1350001	1.14	1.14		
5	1.145	1.14	1.15		
6	1.155	1.16	1.16		

图 2 2 种取舍方法保留 2 位小数的取舍示例

函数式解释:

第一部分为 ROUND(A2, n)。即将 A2 单元格数据按保留 n 位小数进行常规 Excel“四舍五入”法取舍。如图 2 例中各行原始数据常规 Excel“四舍五入”为第三列结果。

第二部分为 $(MOD(A2 * 10^n, 2) = 0.5) / 10^n$ 。MOD(A2 * 10^n, 2) 表示先求 A2 单元格数据乘以 10 的 n 次方再除以 2 的余数。如图 2 例中各行原始数据求余数结果分别为 1.41、0.62、1.500001、0.5、1.5。(MOD(A2 * 10^n, 2) = 0.5) 表示判断上一步求余数结果是不是等于 0.5, 如果等于 0.5 返回 1, 否则返回 0。如图 2 例中各行原始数据仅 1.145 返回值为 1, 其余为 0。(MOD(A2 * 10^n, 2) = 0.5) / 10^n 表示将上一步返回结果除以 10 的 n 次方。如图 2

表 1 2 种取舍方法的差异分析

取用精度位数后一位数字情况	水文流量“四舍六入”法	Excel“四舍五入”法	差异
小于五	舍	舍	相同
大于五	入	入	相同
等于五, 其后有非零尾	入	入	相同
等于五, 其后无非零尾	取用的末位数字为奇	入	相同
	取用的末位数字为偶	舍	相同

例中各行原始数据仅 1.145 返回值为 0.01, 其余为 0。所以第二部分表示仅当取用精度位数后一位数字等于 5, 且其后无非零尾数, 且取用的末位数字为偶数时, 第二部分返回结果为 0.01, 其余为 0。

最后将第一部分结果减去第二部分结果得到最终结果。如图 2 例中各行原始数据仅 1.145 返回值为第一部分结果减去 0.01, 其余为第一部分结果。表示仅当取用精度位数后一位数字等于 5, 且其后无非零尾数, 且取用的末位数字为偶数时, 最终结果会将第一部分结果减去 0.01, 即将第一部分结果中五入的 0.01 舍去, 其余情况采用第一部分结果。从而得到了规范要求的水文流量“四舍六入”取舍方法。

3 使用 Excel 函数实现流量计算中各分步计算

3.1 采用 Excel 函数

IF 函数, 关于这 2 个函数的介绍和用法可以参看本文参考文献^[3]或者 Excel 软件的帮助文档。

3.2 水深计算

规范文件《水文资料整编规范》(SL247-2012)对水深的取用精度方式进行了规范^[2]。“不小于 100 m, 记至 1 m; 小于 100 m, 不小于 5 m, 记至 0.1 m; 小于 5 m, 记至 0.01 m”。所以对水深计算结果采用以下函数式实现精度取舍。

$$=IF(B2 < 5, ROUND(B2, 2) - (MOD(B2 * 10^2, 2) = 0.5) / 10^2, IF(B2 < 100, ROUND(B2, 1) - (MOD(B2 * 10, 2) = 0.5) / 10, ROUND(B2, 0) - (MOD(B2, 2) = 0.5))) \quad (2)$$

3.3 断面面积计算

规范文件《水文资料整编规范》(SL247-2012)对断面面积的取用精度方式进行了规范。“取三位有效数字, 小数不过二位”。可以表示为“断面面积数值不小于 100, 记至 1; 小于 100, 不小于 10, 记至 0.1; 小于 10, 记至 0.01” (由于笔者所在水文站断面面积一般均小于 1000 m², 故不考虑断面面积不小于 1000 m² 的情况)。所以对断面面积计算结果采用以下函数式实现精度取舍。

$$=IF(B5 < 10, ROUND(B5, 2) - (MOD(B5 * 10^2, 2) = 0.5) / 10^2, IF(B5 < 100, ROUND(B5, 1) - (MOD(B5 * 10, 2) = 0.5) / 10, ROUND(B5, 0) - (MOD(B5,$$

$$2) = 0.5))) \quad (3)$$

3.4 流速计算

规范文件《水文资料整编规范》(SL247-2012)对流速的取用精度方式进行了规范。“不小于 1 m/s, 取三位有效数字; 小于 1 m/s, 取二位有效数字, 小数不过三位”。可以表示为“流速数值不小于 10, 记至 0.1; 小于 10, 不小于 0.1, 记至 0.01; 小于 0.1, 记至 0.001” (由于笔者所在水文站流速一般均小于 100 m/s, 故不考虑流速不小于 100 m/s 的情况)。所以对流速计算结果采用以下函数式实现精度取舍。

$$=IF(B9 < 0.1, ROUND(B9, 3) - (MOD(B9 * 10^3, 2) = 0.5) / 10^3, IF(B9 < 10, ROUND(B9, 2) - (MOD(B9 * 10^2, 2) = 0.5) / 10^2, ROUND(B9, 1) - (MOD(B9 * 10, 2) = 0.5) / 10)) \quad (4)$$

3.5 流量计算

规范文件《水文资料整编规范》(SL247-2012)对流量的取用精度方式进行了规范。“取三位有效数字, 小数不过三位”。可以表示为“流量数值不小于 10000, 记至 100; 小于 10000, 不小于 1000, 记至 10; 小于 1000, 不小于 100, 记至 1; 小于 100, 不小于 1, 记至 0.1; 小于 10, 不小于 1, 记至 0.01; 小于 1, 记至 0.001” (由于笔者所在水文站流速一般均小于 100000 m³/s, 故不考虑流速不小于 100000 m³/s 的情况)。所以对流量计算结果采用以下函数式实现精度取舍。

$$=IF(B13 < 1, ROUND(B13, 3) - (MOD(B13 * 10^3, 2) = 0.5) / 10^3, IF(B13 < 10, ROUND(B13, 2) - (MOD(B13 * 10^2, 2) = 0.5) / 10^2, IF(B13 < 100, ROUND(B13, 1) - (MOD(B13 * 10, 2) = 0.5) / 10, IF(B13 < 1000, ROUND(B13, 0) - (MOD(B13, 2) = 0.5), IF(B13 < 10000, ROUND(B13, -1) - (MOD(B13 * 10^-1, 2) = 0.5) / 10^-1, ROUND(B13, -2) - (MOD(B13 * 10^-2, 2) = 0.5) / 10^-2)))) \quad (5)$$

流量计算中各分步计算示例如图 3。

流量计算中各分步计算函数式(2)(3)(4)(5)看似很长和复杂, 其实均为 IF 函数的嵌套使用, 中间全部是插入的函数式(1), 所以各个函数式都很简单易用。将各个函数式用于实测流量计算表中, 通过对同类计算函数式的复制粘贴就可以对流量中各行列数据进行精度取舍, 最终计算出符合规范要求的实测流量值。

C2						=IF(B2<5, ROUND(B2, 2)-(MOD(B2*10^2, 2)=0.5)/10^2, IF(B2<100, ROUND(B2, 1)-(MOD(B2*10, 2)=0.5)/10, ROUND(B2, 0)-(MOD(B2, 2)=0.5)))			
	A	B							
1	数据类别	原始数据1	取舍结果1	原始数据2	取舍结果2				
2	水深	1.125	1.12	112.5	112				
3		8.25	8.2	112.55	113				
4	断面面积	0.225	0.22	222.5	222				
5		2.225	2.22	222.55	223				
6		22.25	22.2	225.5	226				
7	流速	0.0325	0.032	33.25	33.2				
8		0.325	0.32	33.255	33.3				
9		3.325	3.32	33.55	33.6				
10	流量	0.0425	0.042	442.5	442				
11		0.4425	0.442	4425	4420				
12		4.425	4.42	44250	44200				
13		44.25	44.2	44550	44600				

图3 流量计算中各分步计算示例

4 使用 Excel 实现流量计算可能出现问题的探讨

由于 Excel 中浮点运算可能会给出不准确的结果^[4], 本文所采用的计算函数式进行精度取舍后偶尔还是会出现偏差。原因是 Excel 中的数值均是以二进制存储, 而浮点数 (即小数) 以二进制存储时, 大多数为无限循环数, 而且 Excel 中只能存储 15 位精度之内的浮点数, 所以部分小数计算结果就存在尾数取舍误差。比如计算结果是 0.5, 可能在有些情况下就存储为 0.5000000000000028, 而 1.5 有些情况下却存储为 1.499999999999999。在这种情况下出现时本文函数式 (1) 的计算结果就会出现偏差。这种情况出现时, 函数式 (1) 中的 $\text{MOD}(A2*10^n, 2)$ 就会出现本该等于 0.5 的却由于 Excel 中浮点运算误差而不等于 0.5, 故笔者考虑将 $\text{MOD}(A2*10^n, 2)$ 改为 $\text{ROUND}(\text{MOD}(A2*10^n, 2), 10)$, 这样对 $\text{MOD}(A2*10^n, 2)$ 的结果进行小数 10 位的取舍, 消除像 0.5000000000000028 这种 Excel 中浮点运算误差。但这样做也会导致本来计算结果真的为 1.499999999999999 的数据出现精度取舍偏差。但笔者以为后面这种偏差出现的几率比 Excel 中浮点运算误差引起偏差的几率更低 (关于哪种情况的几率更低笔者还未能进行证明, 希望各位读者朋友能够帮忙证明, 或者找到更好的方法), 故笔者采用这种改良过的函数式 (1)。即如下函数式 (6), 用以替代函数式 (2) (3) (4) (5) 中包含的函数式 (1)。

$$=\text{ROUND}(A2, n) - (\text{ROUND}(\text{MOD}(A2*10^n, 2), 10) = 0.5) / 10^n \quad (6)$$

5 结论

本文仅仅选用 ROUND 函数和 MOD 函数 2 个 Excel 函数就实现了符合规范要求的水文流量“四舍六入”取舍方法, 方法简单易学。加上 IF 函数仅仅 3 个 Excel 函数就解决了符合规范要求的水文实测流量计算中水深、断面面积、流速、流量各个分步计算的取用精度取舍问题。从而大大地简化了实测流量的计算过程, 节约了实测流量计算时间。方法简单易学, 便于推广, 从而能提高水文流量测验的工作效率。本文的取用精度取舍方法已经在邛崃水文站的实测流量计算中进行大量运用, 计算结果准确可靠, 得到了同事们的认可。当然, 水平有限, 本文旨在抛砖引玉, 希望广大水文技术人员能从中得到启发, 编写出更多方便实用的水文电算化程序, 从而提高各种水文技术工作的效率。

参考文献:

- [1] GB/T 8170-2008, 数值修约规则与极限数值的表示和判定 [S].
- [2] SL 247-2012, 水文资料整编规范 [S].
- [3] 王作鹏. Excel 2003 从入门到精通 [M]. 北京: 人民邮电出版社, 2013.02.
- [4] 微软. 在 Excel 中浮点运算可能会给出不准确的结果 [EB/OL]. <https://support.microsoft.com/zh-cn/help/78113/floating-point-arithmetic-may-give-inaccurate-results-in-excel#%2Fzh-cn%2Fkb%2F78113>.